

## Modèles SPICE : Aperçu des possibilités

*Auteur : Gérald HUGUENIN, HE-Arc, Baptiste-Savoie 26, CH-2610 St-Imier  
www.he-arc.ch*

1.	Description .....	2
2.	Syntaxe .....	2
3.	Résistance .....	2
4.	Capacité .....	2
5.	MOS .....	3
6.	BIPOLAIRE .....	3
7.	DIODE .....	3
8.	INTERRUPTEUR .....	3
9.	A-O .....	4
9.1.	Modèle idéal (gain fini) .....	4
9.2.	Modèle idéal à un pôle .....	4
9.3.	Modèle complet (extrait de EVAL.LIB de MICROSIM) .....	4
10.	Annexes .....	6
10.1.	Désignation des éléments dans la « netlist » .....	6
10.2.	Types des principaux modèles analogiques .....	7

## 1. Description

On trouvera ci-dessous quelques modèles d'éléments utilisés par le simulateur SPICE et sa version PC (PSPICE). Les modèles doivent être écrits en format texte dans un fichier \*.LIB. Sous PSPICE, ne pas oublier de définir cette librairie dans la liste de bibliothèques à utiliser lors de la simulation. Lors de la première utilisation de la nouvelle librairie, un fichier \*.IND est généré par PSPICE, il faut donc veiller à ce que le disque ne soit pas protégé en écriture.

## 2. Syntaxe

Du type :

```
.model {nom du modèle} {modèle SPICE}(paramètres)
```

Notes : les termes entre crochets {} correspondent à un nom  
les paramètres sont séparés par des virgules  
les nouvelles lignes de paramètres commencent par un signe +  
les commentaires commencent par un signe \*  
plusieurs modèles peuvent être sauvegardés dans un même fichier librairie (\*.LIB)

Exemple :

```
* Library for actives filters  
*  
*EISI G. Huguenin B.-Savoie 26 2610 ST-IMIER  
*  
*                               GH/7.04.93  
*Résistance  
.model Rf RES(R=1 dev=5%)  
*Capacité  
.model Cf CAP(C=1 dev=5%)
```

## 3. Résistance

```
.model Rf RES(R=1 dev=5%)
```

Rf correspond au nom du modèle (doit être identique à celui attribué au symbole)  
RES correspond au nom réservé par SPICE pour un modèle de résistance  
R=x correspond à un facteur de multiplication (1 dans le cas présent)  
dev=x% correspond à tolérance de la résistance (utilisée dans les analyses de Monte-Carlo)  
d'autres paramètres concernant la température peuvent être introduits (cf. p 175 [1])

## 4. Capacité

```
.model Cf CAP(C=1 dev=5%)
```

Cf correspond au nom du modèle (doit être identique à celui attribué au symbole)  
CAP correspond au nom réservé par SPICE pour un modèle de capacité  
C=x correspond à un facteur de multiplication (1 dans le cas présent)  
dev=x% correspond à tolérance de la résistance (utilisée dans les analyses de Monte-Carlo)  
d'autres paramètres concernant la température et la dépendance en tension peuvent être introduits (cf. p 117 [1])

## 5. MOS

\*NMOS array

```
.model NMOSa nmos LEVEL=3 CGSO=2.68E-10 CGDO=2.68E-10 CGBO=0
+ IS=0 FC=0.5 RD=0 RS=0 PHI=0.6 GAMMA=0
+ PB=0.8 CJ=0.000195 CJSW=5E-10 MJ=0.29 MJSW=0.29
+ JS=2.00E-05 RSH=50
+ THETA=3.30E-02 TOX=4.00E-08 dev 20% VTO=0.78 dev 5mV UO=620
+ KAPPA=0.4 VMAX=5.10E+04 ETA=0.138 DELTA=1.4 NFS=9.90E+09
+ NSUB=8.00E+15 XJ=1.50E-07 LD=1.50E-07 AF=1 KF=1.13E-23
+ NSS=4.95E+10 DL=0 DW=0
```

NMOSa correspond au nom du modèle

nmos correspond au nom du modèle de NMOS utilisé par SPICE (pmos est réservé pour le PMOS)

LEVEL =x correspond au niveau de modèle utilisé (sur PSPICE les niveaux 1 à 3 sont acceptés)

Les paramètres débutant par C décrivent les capacités parasites

TOX correspond à l'épaisseur de l'oxyde de grille

VTO correspond à la tension de seuil

Af et Kf correspondent aux paramètres pour l'analyse de bruit

Pour plus de détails se référer à la page 156 de [1])

## 6. BIPOLAIRE

```
.model PNP PNP(IS=4.7e-16 BF=300 NF=.99 VA=20 EG=0.858 XTI=13.38
+ RE=1100)
```

PNP correspond au nom du modèle

PNP correspond au nom du modèle du PNP utilisé par SPICE (npn est réservé pour le npn)

IS correspond au courant inverse de saturation

BF correspond au gain en courant

NF correspond au coefficient d'idéalité

VA correspond à la tension d'Early

EG correspond à la tension de « gap » du Si

XTI correspond à un coefficient de dépendance en température

RE correspond à la résistance d'émetteur

D'autres paramètres peuvent être introduits, pour plus de détails se référer à la page 167 de [1])

## 7. DIODE

```
.model dx D(Is=800.0E-18 Rs=1)
```

DX correspond au nom du modèle

D correspond au nom du modèle de PNP utilisé par SPICE (npn est réservé pour le npn)

IS correspond au courant inverse de saturation

RS correspond à la résistance parasite

D'autres paramètres peuvent être introduits, pour plus de détails se référer à la page 119 de [1])

## 8. INTERRUPTEUR

```
* Library for ideal switches*
```

```
*EISI G. Huguenin B.-Savoie 26 2610 ST-IMIER
```

```
*
```

```
*
```

```
GH/9.98
```

```
*Switch ideal
```

```
.model inter VSWITCH(ROFF=100k VON=5 VOFF=0)
```

**Modélisation SPICE**

inter correspond au nom du modèle

VSWITCH correspond au nom du modèle d'un interrupteur commandé en tension utilisé par SPICE

RON correspond à la résistance lorsque l'interrupteur est en conduction

ROFF correspond à la résistance lorsque l'interrupteur est ouvert

VON correspond à la tension de commande nécessaire pour mettre l'interrupteur en conduction

VOFF correspond à la tension de commande nécessaire pour mettre l'interrupteur à l'état ouvert

Note : Cet élément est utile pour simuler des MOS en commutation en les considérant idéaux.

**9. A-O****9.1. Modèle idéal (gain fini)**

Définir l'A-O dans l'éditeur de schéma à l'aide d'une source de tension dépendante de la tension d'entrée. Le gain est fixé selon les besoins ( $10^5$  pour une valeur pratique). Cf. exemple sous FILTRES.SLB.

**9.2. Modèle idéal à un pôle**

```
* Modèle idéal de l'A-O
*
*
* Gain fini dépendant de f (1 pôle)
*
* Source de tension commandée idéale
*
*
* connections:   non-inverting input
*                | inverting input
*                | | output
*                | | |
.subckt AOideal 1 2 3
*
  Eamp 3 0 LAPLACE { V(1,2) } = { 100000/(1+s/(2*3.14*10)) }
  r1      0 3 1G
.ends
```

Eamp correspond au nom de la source commandée en tension

LAPLACE indique que la source dépend de la fréquence selon l'expression :

$100000/(1+s/(2*3.14*10))$ , la fréquence de coupure est définie à 10 Hz.

La tension de commande est appliquée aux bornes 1 et 2 du bloc

r1 correspond à la résistance entre les deux bornes d'entrées (fixée à 1 GΩ dans ce cas)

**9.3. Modèle complet (extrait de EVAL.LIB de MICROSIM)**

```
* Library of linear IC definitions

* This is a reduced version of MicroSim's linear subcircuit library.
* You are welcome to make as many copies of it as you find convenient.
*
* The parameters in the opamp library were derived from the data sheets for
* each part. The macromodel used is similar to the one described in:
*
* Macromodeling of Integrated Circuit Operational Amplifiers
* by Graeme Boyle, Barry Cohn, Donald Pederson, and James Solomon
* IEEE Journal of Solid-State Circuits, Vol. SC-9, no. 6, Dec. 1974
*
```

```

* Differences from the reference (above) occur in the output limiting stage
* which was modified to reduce internally generated currents associated
with
* output voltage limiting, as well as short-circuit current limiting.
*
* The opamps are modelled at room temperature and do not track changes with
* temperature. This library file contains models for nominal, not worst
case,
* devices.
*
*$
*-----
---
* connections:  non-inverting input
*               | inverting input
*               | | positive power supply
*               | | | negative power supply
*               | | | | output
*               | | | | |
.subckt LM324  1 2 3 4 5
*
  c1  11 12 2.887E-12
  c2   6  7 30.00E-12
  dc   5 53 dx
  de  54  5 dx
  dlp 90 91 dx
  dln 92 90 dx
  dp   4  3 dx
  egnd 99  0 poly(2) (3,0) (4,0) 0 .5 .5
  fb   7 99 poly(5) vb vc ve vlp vln 0 21.22E6 -20E6 20E6 20E6 -20E6
  ga   6  0 11 12 188.5E-6
  gcm  0  6 10 99 59.61E-9
  iee  3 10 dc 15.09E-6
  hlim 90  0 vlim 1K
  q1  11  2 13 qx
  q2  12  1 14 qx
  r2   6  9 100.0E3
  rc1  4 11 5.305E3
  rc2  4 12 5.305E3
  re1 13 10 1.845E3
  re2 14 10 1.845E3
  ree 10 99 13.25E6
  ro1  8  5 50
  ro2  7 99 25
  rp   3  4 9.082E3
  vb   9  0 dc 0
  vc   3 53 dc 1.500
  ve  54  4 dc 0.65
  vlim 7  8 dc 0
  vlp 91  0 dc 40
  vln  0 92 dc 40
.model dx D(Is=800.0E-18 Rs=1)
.model qx PNP(Is=800.0E-18 Bf=166.7)
.ends

```

Ce modèle tient compte de la saturation des étages de sortie, de la limitation en courant. Pour l'étage d'entrée, la paire différentielle est simulée de façon complète (q1 et q2).

Remarque importante : ce modèle ne convient pas pour modéliser un A-O à entrée JFET ou MOS. Dans ce cas, pour une première approximation, on peut augmenter la valeur de BF des PNP d'un facteur 1'000 par exemple.

**10. Annexes**

10.1. Désignation des éléments dans la « netlist »

Désignation	Description
Bxxx	GaAsFET
Cxxx	Capacité
Dxxx	Diode
Exxx	Source de tension commandée en tension
Fxxx	Source de courant commandée en courant
Gxxx	Source de courant commandée en tension
Hxxx	Source de tension commandée en courant
Ixxx	Source de courant
Jxxx	Transistor JFET
Kxxx	Couplage inductif
Lxxx	Inductance
Mxxx	Transistor MOS
Qxxx	Transistor bipolaire
Rxxx	Résistance
Sxxx	Interrupteur commandé en tension
Txxx	Ligne de transmission
Vxxx	Source de tension
Wxxx	Interrupteur commandé en courant
Xxxx	Sous-circuit

avec xxx qui correspond au numéro de l'élément attribué par l'éditeur de schéma

La « netlist » présente des instructions de la forme :

L\_12 1 2 20mH (inductance de 20mH connectée entre les noeuds 1 et 2)

Certains de ces éléments peuvent ou doivent être complétés par la définition d'un *modèle* dans une librairie (\*.lib) selon la liste suivante :

10.2. Types des principaux modèles analogiques

Type du modèle	Désignation dans la « netlist »	Description
CAP	Cxxx	Capacité
CORE	Kxxx	Transformateur
D	Dxxx	Diode
GASFET	Bxxx	Transistor FET au GaAs
IND	Lxxx	Inductance
ISWITCH	Wxxx	Interrupteur commandé en courant
NJF	Jxxx	Transistor JFET canal n
NMOS	Mxxx	Transistor MOS canal n
NPN	Qxxx	Transistor bipolaire NPN
PJF	Jxxx	Transistor JFET canal p
PMOS	Mxxx	Transistor MOS canal p
PNP	Qxxx	Transistor PNP
RES	Rxxx	Résistance
VSWITCH	Sxxx	Interrupteur commandé en tension

avec xxx qui correspond au numéro de l'élément attribué par l'éditeur de schéma

La librairie doit contenir une description du type :

```
.model {nom du modèle} {type de modèle} (paramètres)
```

avec :

```
{nom du modèle} : nom du modèle (au choix de l'utilisateur)
```

```
{type de modèle} : type de modèle selon la liste ci-dessus
```

```
paramètres : paramètres propres à chaque modèle
```